

DESIGN NOTES

Number 12 in a series from Linear Technology Corporation

July, 1988

An LT1013 and LT1014 Op Amp SPICE Macromodel

Walter G. Jung

With the advent of low cost and powerful desktop computers, present day op amp circuit designs can mature more quickly with good simulation tools. One such tool since its inception has been SPICE, the standard analog circuit simulator. However, while PCs and workstations may now be present on more and more desks, a potential bottleneck towards effective simulation has been SPICE models for the more popular parts.

The macromodel approach to simulation of an op amp is viable for many designs, with the great asset of simulation speeds far faster than that of a full device-level circuit. With this design note, Linear Technology Corporation introduces

op amp macromodels to its applications library. It is hoped that eventually most op amps in the product line will be developed as macromodels and made available to customers.

The LT1013 and LT1014 devices are popular single supply LTC op amps, and are thus logical candidates for macromodels. While existing macromodels for the generic 358 and 324 types might suffice for some applications, circuit designs which take advantage of the unique precision and functional features of the LT1013 warrant a model which reflects those features. The schematic diagram of the LT1013 and LT1014 macromodel is shown in Figure 1, and is applicable to one channel of either device.




Figure 1. LT1013 Op Amp Macromodel

Key op-amp specifications for the commercial device are:
 offset voltage = $50\mu\text{V}$ (offset is not simulated)
 bias current = 8nA
 gain = 1000000
 slew rate = $0.4\text{V}/\mu\text{s}$
 bandwidth = 0.8MHz

Also, the model simulates the input common mode range (which includes ground) and the output characteristics of swinging to ground while sinking current.

This macromodel acts very much like the real LT1013 or LT1014 device which incorporates input common mode clamping, to prevent the sign-reversal errors common to the 358/324 types. For example, comparison responses for the LT1013 and the 358 are shown in Figure 2. The common conditions of this test are for an overdriven, +5V single-supply follower. In both instances, the input signal is V_{IN} , a -20V to 20V sweep fed through 10k , while the output is $V(5)$.

Note that with the 358, the output reverses sign when the input is overdriven below ground. In contrast, the LT1013 model is well behaved, simply clamping the overdrive at ground level...just like the real LT1013 device does!

The model itself is listed on this page, and can be entered by typing it in (carefully!). Registered users of MicroSim's PSPICE simulator will automatically receive this macromodel as part of the model library update with version 3.07. Interested readers may contact MicroSim at the address or phone number listed at the end of this note for further information.

This LT1013/LT1014 op amp macromodel is being supplied to users as an aid to circuit designs. While it reflects reasonably close similarity to the actual device in terms of performance, it is not suggested as a replacement for breadboarding. Simulation should be used as a forerunner or a supplement to traditional lab testing.




Figure 2. LT1013 Test Circuit: Single-Supply (+ 5V), Overdriven Follower

This more complete macromodel has been adapted from the Parts program generated LT1013/LT1014 model. This version features closer fidelity to the real part, with input common mode clamping and compensated output clamping. It can be used for large signal and/or single supply applications, where the inputs can potentially be overdriven.

SPICE List for LT1013 Macromodel

```
connections:
*      | non-inverting input
*      | | inverting input
*      | | | positive power supply
*      | | | | negative power supply
*      | | | | | output
*      | | | | | |
.subckt LT1013 1 2 3 4 5
*
*
c1      11 12      8.661E-12
c2      6 7      30.00E-12
dc      8 53 dx
de      54 8 dx
dlp     90 91 dx
dln     92 90 dx
dp      4 3 dx
egnd    99 0      poly(2) (3,0) (4,0) 0.5 .5
fb      7 99      poly(5) vb vc ve vip vln 0
          2.475E9 -2E9 2E9 2E9 -2E9
ga      6 0 11 12 113.1E-6
gcm     0 6 10 99 225.7E-12
iee     3 10 dc 12.03E-6
hlim    90 0      vlim 1k
q1      11 102 13 qx
q2      12 101 14 qx
rb1     2 102 400
rb2     1 101 400
dcm1    105 102 dx
dcm2    105 101 dx
vcmc    105 4 dc 0.4
r2      6 9      100.0E3
rc1     4 11      8.841E3
rc2     4 12      8.841E3
re1     13 10     4.519E3
re2     14 10     4.519E3
ree     10 99     16.63E6
ro1     8 5 80
ro2     7 99 25
ip      3 4      328E-6; supply current
vb      9 0 dc 0
vc      3 53 dc 1.610
ve      54 4 dc .61
vlim    7 8 dc 0
vlp     91 0 dc 25
vln     0 92 dc 25
.model dx D(ls = 800.0E-18)
.model qx PNP(ls = 800.0E-18 Bf = 400)
.ends

PSPICE simulator is available from:
MicroSim
23175 La Cadena Drive
Laguna Hills, CA 92653
(714) 770-3022; (800) 826-8603
```

For LT1013/14 literature call 800-637-5545. For help with an application call (408) 432-1900, Ext. 361.